

# Chapter 24

## Cyclohexane

### 24.1 Statement of the Problem


Cyclohexane molecule contains 6 C atoms connected cyclically by single chemical bonds; 2 H atoms are attached to each carbon one. Single bonds of a C atom have a fixed length and form fixed angles: if the C atom is put to the center of a tetrahedron, then its single bonds point to its vertices. The problem is: how many geometrical configurations (conformations, as chemists call them) of the cyclohexane molecule exist—one, several, or infinitely many (and if so, what is the dimensionality of this set).

#### *Tetrahedron*

What is the angle between single bonds? The unit vectors  $a[1], \dots, a[4]$  (blue) are directed from the center of the tetrahedron (red) to its vertices.

```
In[1] := a[0] = {0, 0, 0}; a[1] = {0, 0, 1}; a[2] = {2 * Sqrt[2], 0, -1}/3;
a[3] = {-Sqrt[2], Sqrt[6], -1}/3; a[4] = {-Sqrt[2], -Sqrt[6], -1}/3;
In[2] := rc = 0.025; rs = 0.1;
In[3] := Graphics3D[{Blue, Cylinder[{a[0], a[1]}, rc], Cylinder[{a[0], a[2]}, rc],
Cylinder[{a[0], a[3]}, rc], Cylinder[{a[0], a[4]}, rc],
Red, Cylinder[{a[1], a[2]}, rc], Cylinder[{a[1], a[3]}, rc],
Cylinder[{a[1], a[4]}, rc], Cylinder[{a[2], a[3]}, rc],
Cylinder[{a[2], a[4]}, rc], Cylinder[{a[3], a[4]}, rc],
Sphere[a[1], rs], Sphere[a[2], rs], Sphere[a[3], rs], Sphere[a[4], rs]},
Boxed->False, ViewPoint->{10, 10, 4}]
```

Out[3] =



Let's check: the vectors are indeed unit and form equal angles with each other; hence all edges have equal lengths, so this is indeed a tetrahedron.

```
In[4] := MatrixForm[Table[a[i].a[j], {i, 1, 4}, {j, 1, 4}]]
```

```
Out[4] // MatrixForm =
```

$$\begin{pmatrix} 1 & -\frac{1}{3} & -\frac{1}{3} & -\frac{1}{3} \\ -\frac{1}{3} & 1 & -\frac{1}{3} & -\frac{1}{3} \\ -\frac{1}{3} & -\frac{1}{3} & 1 & -\frac{1}{3} \\ -\frac{1}{3} & -\frac{1}{3} & -\frac{1}{3} & 1 \end{pmatrix}$$

Cosine of the angle between these vectors is  $-1/3$ .

```
In[5] := Clear[a, rc, rs]
```

## 24.2 First Steps

Thus we have 6 vectors  $a[1], \dots, a[6]$  drawn from each C atom to the next one. They form a closed hexagon:  $a[1] + \dots + a[6] = 0$ . Let the length of a single C–C bond be 1, then all the vectors are unit. Scalar products of neighboring vectors are  $1/3$  (the sign has changed because now one of the vectors points in, not out). In order not to deal with the overall orientation of the molecule, let's consider invariant quantities—cosines of the angles  $c[i, j] = a[i].a[j]$ .

```
In[6] := c[i_-, j_-] /; i > j := c[j, i]
```

```
In[7] := Do[c[i, i] = 1, {i, 1, 6}]
```

```
In[8] := Do[c[i, i + 1] = 1/3, {i, 1, 5}]; c[1, 6] = 1/3;
```

**In[9] := MatrixForm[M = Array[c, {6, 6}]]**

Out[9]//MatrixForm =

$$\begin{pmatrix} 1 & \frac{1}{3} & c[1,3] & c[1,4] & c[1,5] & \frac{1}{3} \\ \frac{1}{3} & 1 & \frac{1}{3} & c[2,4] & c[2,5] & c[2,6] \\ c[1,3] & \frac{1}{3} & 1 & \frac{1}{3} & c[3,5] & c[3,6] \\ c[1,4] & c[2,4] & \frac{1}{3} & 1 & \frac{1}{3} & c[4,6] \\ c[1,5] & c[2,5] & c[3,5] & \frac{1}{3} & 1 & \frac{1}{3} \\ \frac{1}{3} & c[2,6] & c[3,6] & c[4,6] & \frac{1}{3} & 1 \end{pmatrix}$$

## Linear Equations

Multiplying the vector equality  $a[1] + \dots + a[6] = 0$  by each vector  $a[i]$ , we get 6 linear equations for  $c[i, j]$ .

**In[10] := Eq = Table[Sum[c[i, j], {j, 1, 6}] == 0, {i, 1, 6}]**

$$\text{Out[10]} = \left\{ \begin{aligned} \frac{5}{3} + c[1,3] + c[1,4] + c[1,5] &== 0, \frac{5}{3} + c[2,4] + c[2,5] + c[2,6] == 0, \\ \frac{5}{3} + c[1,3] + c[3,5] + c[3,6] &== 0, \frac{5}{3} + c[1,4] + c[2,4] + c[4,6] == 0, \\ \frac{5}{3} + c[1,5] + c[2,5] + c[3,5] &== 0, \frac{5}{3} + c[2,6] + c[3,6] + c[4,6] == 0 \end{aligned} \right\}$$

Let's take  $x = c[1,3]$ ,  $y = c[3,5]$ ,  $z = c[5,1]$  as independent variables and express the remaining ones via them.

**In[11] := c[1,3] = x; c[3,5] = y; c[1,5] = z;**

**In[12] := s = Solve[Eq, {c[1,4], c[2,4], c[2,5], c[2,6], c[3,6], c[4,6]}][[1]]**

$$\text{Out[12]} = \left\{ \begin{aligned} c[1,4] \rightarrow -\frac{5}{3} - x - z, c[2,4] \rightarrow z, c[2,5] \rightarrow -\frac{5}{3} - y - z, c[2,6] \rightarrow y, \\ c[3,6] \rightarrow -\frac{5}{3} - x - y, c[4,6] \rightarrow x \end{aligned} \right\}$$

**In[13] := c[1,4] = c[1,4]/.s; c[2,4] = c[2,4]/.s; c[2,5] = c[2,5]/.s;**

**c[2,6] = c[2,6]/.s; c[3,6] = c[3,6]/.s; c[4,6] = c[4,6]/.s;**

**In[14] := Clear[s]**

**In[15] := MatrixForm[M = M]**

Out[15]//MatrixForm =

$$\begin{pmatrix} 1 & \frac{1}{3} & x & -\frac{5}{3} - x - z & z & \frac{1}{3} \\ \frac{1}{3} & 1 & \frac{1}{3} & z & -\frac{5}{3} - y - z & y \\ x & \frac{1}{3} & 1 & \frac{1}{3} & y & -\frac{5}{3} - x - y \\ -\frac{5}{3} - x - z & z & \frac{1}{3} & 1 & \frac{1}{3} & x \\ z & -\frac{5}{3} - y - z & y & \frac{1}{3} & 1 & \frac{1}{3} \\ \frac{1}{3} & y & -\frac{5}{3} - x - y & x & \frac{1}{3} & 1 \end{pmatrix}$$

## 24.3 Equations

### *Generating Combinations*

Now let's recall that our vectors  $a[1], \dots, a[6]$  live in 3-dimensional space. Any 4 of them are linearly dependent. First we have to find a way to generate all 4-element lists made of the numbers from 1 to 6 in the increasing order. We shall accumulate them in the list  $L$ . The main work is done by the recursive function  $\text{Gen}$ . Its parameters:  $l$ —part of the list which has been already constructed;  $n$ —how many elements are to be added;  $a$  and  $b$ —boundaries of the interval from which numbers can be taken. If everything has been done ( $n = 0$ ), the constructed list  $l$  is appended to the list of results  $L$ . Otherwise, we add each number  $i$  from the allowed interval to  $l$  and call  $\text{Gen}$  recursively to add  $n - 1$  numbers. The upper limit of the loop is determined by the requirement to have at least  $n - 1$  numbers in the interval from  $i + 1$  to  $b$ .

**In[16] :=  $L = \{\}$ ;**

**In[17] :=  $\text{Gen}[l \_, n \_, a \_, b \_] := \text{If}[n \leq 0, L = \text{Append}[L, l],$**

**Do[ $\text{Gen}[\text{Append}[l, i], n - 1, i + 1, b], \{i, a, b - n + 1\}]$**

**In[18] :=  $\text{Gen}[\{\}, 4, 1, 6]; L$**

**Out[18] =  $\{\{1, 2, 3, 4\}, \{1, 2, 3, 5\}, \{1, 2, 3, 6\}, \{1, 2, 4, 5\}, \{1, 2, 4, 6\}, \{1, 2, 5, 6\},$**   
 $\{1, 3, 4, 5\}, \{1, 3, 4, 6\}, \{1, 3, 5, 6\}, \{1, 4, 5, 6\}, \{2, 3, 4, 5\}, \{2, 3, 4, 6\},$   
 $\{2, 3, 5, 6\}, \{2, 4, 5, 6\}, \{3, 4, 5, 6\}\}$

It works. There are 15 4-combinations of 6 numbers.

### *Nonlinear Equations*

For each set of 4 vectors, the determinant of the corresponding  $4 \times 4$  submatrix of the matrix  $M$  (it is the square of the 4-dimensional volume spanned by these vectors) should be equal to 0.

**In[19] :=  $\text{Eq} = \text{Table}[\text{Det}[M[[l, l]]], \{l, L\}]$**

**Out[19] =  $\left\{ \begin{array}{l} \frac{5}{3} - \frac{34x}{9} - \frac{23x^2}{9} - \frac{34z}{9} - \frac{20xz}{9} + \frac{2x^2z}{3} - \frac{23z^2}{9} + \frac{2xz^2}{3} + x^2z^2, \\ -2 + \frac{2x}{9} + \frac{16x^2}{9} - \frac{40y}{9} + \frac{10xy}{9} + \frac{10x^2y}{3} - \frac{23y^2}{9} + \frac{2xy^2}{3} + x^2y^2 - \frac{40z}{9} + \\ \frac{10xz}{9} + \frac{10x^2z}{3} - \frac{32yz}{9} + \frac{10xyz}{3} + 2x^2yz - \frac{23z^2}{9} + \frac{2xz^2}{3} + x^2z^2, \\ -\frac{5}{3} - \frac{34x}{9} - \frac{23x^2}{9} - \frac{34y}{9} - \frac{20xy}{9} + \frac{2x^2y}{3} - \frac{23y^2}{9} + \frac{2xy^2}{3} + x^2y^2, \\ \frac{7}{3} + \frac{50x}{9} + \frac{16x^2}{9} + \frac{50y}{9} + \frac{98xy}{9} + \frac{10x^2y}{3} + \frac{16y^2}{9} + \frac{10xy^2}{3} + x^2y^2 + \frac{20z}{3} + \\ \frac{118xz}{9} + \frac{10x^2z}{3} + \frac{118yz}{9} + \frac{40xyz}{3} + 2x^2yz + \frac{10y^2z}{3} + 2xy^2z + 4z^2 + \end{array} \right.$**

$$\begin{aligned}
& \frac{20xz^2}{3} + x^2z^2 + \frac{20yz^2}{3} + 2xyz^2 + y^2z^2, \\
-2 & - \frac{40x}{9} - \frac{23x^2}{9} + \frac{2y}{9} + \frac{10xy}{9} + \frac{2x^2y}{3} + \frac{16y^2}{9} + \frac{10xy^2}{3} + x^2y^2 - \frac{40z}{9} - \\
& \frac{32xz}{9} + \frac{10yz}{9} + \frac{10xyz}{3} + \frac{10y^2z}{3} + 2xy^2z - \frac{23z^2}{9} + \frac{2yz^2}{3} + y^2z^2, \\
- \frac{5}{3} & - \frac{34x}{9} - \frac{23y^2}{9} - \frac{34z}{9} - \frac{20yz}{9} + \frac{2y^2z}{3} - \frac{23z^2}{9} + \frac{2yz^2}{3} + y^2z^2, \\
-2 & - \frac{40x}{9} - \frac{23x^2}{9} + \frac{2y}{9} + \frac{10xy}{9} + \frac{2x^2y}{3} + \frac{16y^2}{9} + \frac{10xy^2}{3} + x^2y^2 - \frac{40z}{9} - \\
& \frac{32xz}{9} + \frac{10yz}{9} + \frac{10xyz}{3} + \frac{10y^2z}{3} + 2xy^2z - \frac{23z^2}{9} + \frac{2yz^2}{3} + y^2z^2, \\
\frac{7}{3} & + \frac{20x}{3} + 4x^2 + \frac{50y}{9} + \frac{118xy}{9} + \frac{20x^2y}{3} + \frac{16y^2}{9} + \frac{10xy^2}{3} + x^2y^2 + \frac{50z}{9} + \\
& \frac{118xz}{9} + \frac{20x^2z}{3} + \frac{98yz}{9} + \frac{40xyz}{3} + 2x^2yz + \frac{10y^2z}{3} + 2xy^2z + \frac{16z^2}{9} + \\
& \frac{10xz^2}{3} + x^2z^2 + \frac{10yz^2}{3} + 2xyz^2 + y^2z^2, \\
-2 & - \frac{40x}{9} - \frac{23x^2}{9} - \frac{40y}{9} - \frac{32xy}{9} - \frac{23y^2}{9} + \frac{2z}{9} + \frac{10xz}{9} + \frac{2x^2z}{3} + \frac{10yz}{9} + \\
& \frac{10xyz}{3} + \frac{2y^2z}{3} + \frac{16z^2}{9} + \frac{10xz^2}{3} + x^2z^2 + \frac{10yz^2}{3} + 2xyz^2 + y^2z^2, \\
- \frac{5}{3} & - \frac{34x}{9} - \frac{23x^2}{9} - \frac{34z}{9} - \frac{20xz}{9} + \frac{2x^2z}{3} - \frac{23z^2}{9} + \frac{2xz^2}{3} + x^2z^2, \\
- \frac{5}{3} & - \frac{34y}{9} - \frac{23y^2}{9} - \frac{34z}{9} - \frac{20yz}{9} + \frac{2y^2z}{3} - \frac{23z^2}{9} + \frac{2yz^2}{3} + y^2z^2, \\
-2 & - \frac{40x}{9} - \frac{23x^2}{9} - \frac{40y}{9} - \frac{32xy}{9} - \frac{23y^2}{9} + \frac{2z}{9} + \frac{10xz}{9} + \frac{2x^2z}{3} + \frac{10yz}{9} + \\
& \frac{10xyz}{3} + \frac{2y^2z}{3} + \frac{16z^2}{9} + \frac{10xz^2}{3} + x^2z^2 + \frac{10yz^2}{3} + 2xyz^2 + y^2z^2, \\
\frac{7}{3} & + \frac{50x}{9} + \frac{16x^2}{9} + \frac{20y}{3} + \frac{118xy}{9} + \frac{10x^2y}{3} + 4y^2 + \frac{20xy^2}{3} + x^2y^2 + \frac{50z}{9} + \\
& \frac{98xz}{9} + \frac{10x^2z}{3} + \frac{118yz}{9} + \frac{40xyz}{3} + 2x^2yz + \frac{20y^2z}{3} + 2xy^2z + \frac{16z^2}{9} + \\
& \frac{10xz^2}{3} + x^2z^2 + \frac{10yz^2}{3} + 2xyz^2 + y^2z^2, \\
-2 & + \frac{2x}{9} + \frac{16x^2}{9} - \frac{40y}{9} + \frac{10xy}{9} + \frac{10x^2y}{3} - \frac{23y^2}{9} + \frac{2xy^2}{3} + x^2y^2 - \frac{40z}{9} + \\
& \frac{10xz}{9} + \frac{10x^2z}{3} - \frac{32yz}{9} + \frac{10xyz}{3} + 2x^2yz - \frac{23z^2}{9} + \frac{2xz^2}{3} + x^2z^2, \\
- \frac{5}{3} & - \frac{34x}{9} - \frac{23x^2}{9} - \frac{34y}{9} - \frac{20xy}{9} + \frac{2x^2y}{3} - \frac{23y^2}{9} + \frac{2xy^2}{3} + x^2y^2 \}
\end{aligned}$$

These 15 polynomials of 3 variables must be equal to 0.

**In[20] := Clear[L]**

## Gröbner Basis

Let's find a simpler set of polynomials having the same solution set—the Gröbner basis.

**In[21] := GB = GroebnerBasis[Eq, {z, y, x}]**

Out[21] =  $\{-15 - 34x - 23x^2 - 34y - 20xy + 6x^2y - 23y^2 + 6xy^2 + 9x^2y^2,$   
 $-102 - 400x - 284x^2 + 18x^4 - 69y - 212xy + 18x^2y + 108x^3y + 27x^4y -$   
 $69z - 212xz + 18x^2z + 108x^3z + 27x^4z,$   
 $18 + 54x + 18x^2 - 6x^3 + 21y + 41xy - 9x^2y - 9x^3y + 21z + 41xz -$   
 $9x^2z - 9x^3z + 20yz,$   
 $-15 - 34x - 23x^2 - 34z - 20xz + 6x^2z - 23z^2 + 6xz^2 + 9x^2z^2\}$

Do they factorize?

**In[22] := GB = Map[Factor, GB]**

Out[22] =  $\{-15 - 34x - 23x^2 - 34y - 20xy + 6x^2y - 23y^2 + 6xy^2 + 9x^2y^2,$   
 $(3 + x)(1 + 3x)(-34 - 20x + 6x^2 - 23y + 6xy + 9x^2y - 23z + 6xz + 9x^2z),$   
 $18 + 54x + 18x^2 - 6x^3 + 21y + 41xy - 9x^2y - 9x^3y + 21z + 41xz - 9x^2z -$   
 $9x^3z + 20yz,$   
 $-15 - 34x - 23x^2 - 34z - 20xz + 6x^2z - 23z^2 + 6xz^2 + 9x^2z^2\}$

**In[23] := p1 = GB[[1]]; p2 = GB[[2]]/(3 + x)/(1 + 3 \* x);**

**p3 = GB[[3]]; p4 = GB[[4]];**

## 24.4 Projection onto the x, y Plane

### Allowed Region

First let's find the projection of the solution set onto the x, y plane. What part of this plane are we interested in? First, x and y should lie between -1 and 1; second,


**In[24] := c[3, 6]**

Out[24] =  $-\frac{5}{3} - x - y$

should also lie between -1 and 1.

**In[25] := RegionPlot[-1 ≤ x ≤ 1 && -1 ≤ y ≤ 1 && -1 ≤ c[3, 6] ≤ 1,**

**{x, -1, 1}, {y, -1, 1}]**



That is, the allowed region is the triangle with the vertices  $(-1, -1)$ ,  $(-1, 1/3)$ , and  $(1/3, -1)$ .

**Solutions with  $x = -1/3$**

The second equation is satisfied at  $x = -1/3$ . What about the other ones?

**In[26] := Eq = {p1, p3, p4} /. x -> -1/3**

$$\text{Out[26]} = \left\{ -\frac{56}{9} - \frac{80y}{3} - 24y^2, \frac{20}{9} + \frac{20y}{3} + \frac{20z}{3} + 20yz, -\frac{56}{9} - \frac{80z}{3} - 24z^2 \right\}$$

**In[27] := GB = GroebnerBasis[Eq, {z, y}]**

$$\text{Out[27]} = \{7 + 30y + 27y^2, 1 + 3y + 3z + 9yz, 7 + 30z + 27z^2\}$$

**In[28] := GB = Map[Factor, GB]**

$$\text{Out[28]} = \{(1 + 3y)(7 + 9y), (1 + 3y)(1 + 3z), (1 + 3z)(7 + 9z)\}$$

So, we have found the solutions  $x = y = z = -1/3$ ;  $x = y = -1/3, z = -7/9$ ; and  $x = z = -1/3, y = -7/9$ .

**In[29] := Eq /. {y -> -1/3, z -> -1/3}**

$$\text{Out[29]} = \{0, 0, 0\}$$

**In[30] := Eq /. {y -> -1/3, z -> -7/9}**

$$\text{Out[30]} = \{0, 0, 0\}$$

**In[31] := Eq/.{y->-7/9,z->-1/3}**

Out[31] = {0,0,0}

It is clear from the symmetry argument that  $y = z = -1/3$ ,  $x = -7/9$  is also a solution.

**In[32] := {p1,p2,p3,p4}/.{x->-7/9,y->-1/3,z->-1/3}**

Out[32] = {0,0,0,0}

### Other Solutions


The first equation contains only  $x$  and  $y$ .

**In[33] := p1**

Out[33] =  $-15 - 34x - 23x^2 - 34y - 20xy + 6x^2y - 23y^2 + 6xy^2 + 9x^2y^2$

**In[34] := P1 = ContourPlot[p1 == 0, {x, -1, 0}, {y, -1, 0}]**

Out[34] =



This equation is quadratic in  $y$ .

**In[35] := Do[c[i] = Coefficient[p1,y,i], {i,0,2}]**

Does  $c[2]$  vanish somewhere?

**In[36] := s = Solve[c[2] == 0, x]**

Out[36] =  $\left\{ \left\{ x \rightarrow \frac{1}{3} (-1 - 2\sqrt{6}) \right\}, \left\{ x \rightarrow \frac{1}{3} (-1 + 2\sqrt{6}) \right\} \right\}$

**In[37] := N[x/.s]**

Out[37] =  $\{-1.96633, 1.29966\}$

In our region  $c[2] > 0$ . The discriminant:

**In[38] := d = Factor[c[1]^2 - 4 \* c[0] \* c[2]]**

Out[38] =  $32(-1+x)(7+9x)(1+6x+3x^2)$

**In[39] := s = Solve[d == 0, x]**

Out[39] =  $\left\{ \left\{ x \rightarrow -\frac{7}{9} \right\}, \{x \rightarrow 1\}, \left\{ x \rightarrow \frac{1}{3}(-3 - \sqrt{6}) \right\}, \left\{ x \rightarrow \frac{1}{3}(-3 + \sqrt{6}) \right\} \right\}$

**In[40] := N[x/.s]**

Out[40] =  $\{-0.777778, 1, -1.8165, -0.183503\}$

The discriminant is positive between  $x = -7/9$  and

**In[41] := xmax = x/.s[[4]]**

Out[41] =  $\frac{1}{3}(-3 + \sqrt{6})$

**In[42] := Clear[s]**

Two values of y (with  $\text{pm} = \pm 1$ ) correspond to each x from this interval.

**In[43] := y1 = (-c[1] + pm \* Sqrt[d]) / (2 \* c[2])**

Out[43] =  $\left( 34 + 20x - 6x^2 + 4\sqrt{2}\text{pm}\sqrt{(-1+x)(7+9x)(1+6x+3x^2)} \right) / (2(-23+6x+9x^2))$

**In[44] := Clear[d]**

Do the points we found earlier lie on this curve?

**In[45] := {p1/.{x->-1/3,y->-1/3}, p1/.{x->-1/3,y->-7/9}, p1/.{x->-7/9,y->-1/3}}**

Out[45] =  $\{0, 0, 0\}$

Yes, they do. Let's denote these points A, B, C.

**In[46] := pA = {-1/3, -1/3}; pB = {-1/3, -7/9}; pC = {-7/9, -1/3};**

So, all solutions we are interested in project onto this curve in the x, y plane.

We shall need a few extra points on this curve. Let's find the second intersection with the diagonal  $x = y$  (the first one is the point A).

**In[47] := s = Solve[(p1/.y->x) == 0, x]**

Out[47] =  $\left\{ \{x \rightarrow -3\}, \left\{ x \rightarrow -\frac{1}{3} \right\}, \left\{ x \rightarrow \frac{1}{3}(3 - 2\sqrt{6}) \right\}, \left\{ x \rightarrow \frac{1}{3}(3 + 2\sqrt{6}) \right\} \right\}$

**In[48] := N[x/.s]**

Out[48] =  $\{-3., -0.333333, -0.632993, 2.63299\}$

**In[49] := x0 = x/.s[[3]]**

Out[49] =  $\frac{1}{3}(3 - 2\sqrt{6})$

Let's call this point D.

**In[50] := pD = {x0, x0};**

**In[51] := Clear[s, x0]**

Finally, let's introduce some additional point between A and B (rather arbitrarily; e.g., let it have  $x = -1/4$ ) and call it E. Let its mirror image be the point F.

```
In[52] := x0 = -1/4;
```

```
In[53] := y0 = Simplify[y1/. {x->x0, pm->1}]
```

```
Out[53] =  $\frac{1}{383} (-229 - 10\sqrt{38})$ 
```


```
In[54] := pE = {x0, y0}; pF = {y0, x0};
```

```
In[55] := Clear[x0, y0]
```

This is shown in the plot.

```
In[56] := P2 = Graphics[{PointSize[Large],
  Red, Point[pA], Text[Style[A, Large], pA, {-1, -1}],
  Point[pB], Text[Style[B, Large], pB, {0, 1}],
  Point[pC], Text[Style[C, Large], pC, {1, 0}],
  Darker[Green], Point[pD], Text[Style[D, Large], pD, {1, 1}],
  Point[pE], Text[Style[E, Large], pE, {-1, 1}],
  Point[pF], Text[Style[F, Large], pF, {1, -1}]}];
```

```
In[57] := Show[P1, P2]
```




What's the reason for introducing the points D, E, F? Now it is easy to write down our curve parametrically. For  $t \in [0, 1]$  let the point move from F to A, the motion along  $x$  being uniform.

```
In[58] := pFA[t_] := With[{xt = (1 - t) * pF[[1]] + t * pA[[1]]},
  {xt, y1/. {x->xt, pm->1}}]
```

```
In[59] := P1 = ParametricPlot[pFA[t], {t, 0, 1}, PlotRange->{{-1, 0}, {-1, 0}}];
```

In[60] := Show[P1,P2]




Out[60] =

For  $t \in [1,2]$  let the point move from A to E; this segment is mirror-symmetric to the previous one.

In[61] := pAE[t\_] := With[{y0 = (2 - t) \* pA[[2]] + (t - 1) \* pE[[2]]},  
 {y1/. {x->y0, pm->-1}, y0}]

In[62] := P1 = ParametricPlot[pAE[t], {t, 1, 2}, PlotRange->{{-1, 0}, {-1, 0}}];

In[63] := Show[P1,P2]



Out[63] =


For  $t \in [2, 3]$  the point moves from E to D, the motion along  $x$  being uniform.

```
In[64] := pED[t_] := With[{x0 = (3 - t) * pE[[1]] + (t - 2) * pD[[1]]},
  {x0, y1 /. {x -> x0, pm -> 1}}]
```

```
In[65] := P1 = ParametricPlot[pED[t], {t, 2, 3}, PlotRange -> {{-1, 0}, {-1, 0}}];
```

```
In[66] := Show[P1, P2]
```

Out[66] =




Finally, for  $t \in [3, 4]$  the point moves from D to F; this segment is mirror-symmetric to the previous one.

```
In[67] := pDF[t_] := With[{y0 = (4 - t) * pD[[2]] + (t - 3) * pF[[2]]},
  {y1 /. {x -> y0, pm -> 1}, y0}]
```

```
In[68] := P1 = ParametricPlot[pDF[t], {t, 3, 4}, PlotRange -> {{-1, 0}, {-1, 0}}];
```

```
In[69] := Show[P1, P2]
```

Out[69] =



Later we shall join these segments and construct a parametric curve in the 3-dimensional space  $x, y, z$ .

## 24.5 Complete Analysis of the Solutions

How to find the value (or values) of  $z$  corresponding to some point  $x, y$  on our curve? It is easiest to use the second equation—it is linear in  $z$ .

**In[70] := p2**

Out[70] =  $-34 - 20x + 6x^2 - 23y + 6xy + 9x^2y - 23z + 6xz + 9x^2z$

**In[71] := Do[c[i] = Coefficient[p2, z, i], {i, 0, 1}]**

Does  $c[1]$  vanish somewhere in our region?

**In[72] := s = Solve[c[1] == 0, x]**

Out[72] =  $\left\{ \left\{ x \rightarrow \frac{1}{3} (-1 - 2\sqrt{6}) \right\}, \left\{ x \rightarrow \frac{1}{3} (-1 + 2\sqrt{6}) \right\} \right\}$

**In[73] := N[x/.s]**

Out[73] =  $\{-1.96633, 1.29966\}$

No, it does not.

**In[74] := Clear[s]**

So there is a single solution:

**In[75] := z1 = -c[0]/c[1]**

Out[75] =  $\frac{34 + 20x - 6x^2 + 23y - 6xy - 9x^2y}{-23 + 6x + 9x^2}$

And what about the third and fourth equations?

**In[76] := p3 = Numerator[Together[p3/.z->z1]]**

Out[76] =  $-20(-15 - 34x - 23x^2 - 34y - 20xy + 6x^2y - 23y^2 + 6xy^2 + 9x^2y^2)$

**In[77] := p4 = Numerator[Together[p4/.z->z1]]**

Out[77] =  $-15 - 34x - 23x^2 - 34y - 20xy + 6x^2y - 23y^2 + 6xy^2 + 9x^2y^2$

**In[78] := Cancel[p3/p1]**

Out[78] =  $-20$

**In[79] := Cancel[p4/p1]**

Out[79] =  $1$

They are satisfied automatically. What  $z$  corresponds to  $x = y = -1/3$ ?

**In[80] := z1 /. {x->-1/3, y->-1/3}**

Out[80] =  $-\frac{7}{9}$

So, one of the solutions found earlier, namely  $x = y = z = -1/3$ , does not belong to our one-dimensional family of solutions. To summarize: we have found one isolated solution plus a one-dimensional family of solutions. In the parametric form:

**In[81] := xyz[t\_] := With[{xy = Which[t < 1, pFA[t], t < 2, pAE[t],  
t < 3, pED[t], True, pDF[t]]},**

**{xy[[1]], xy[[2]], z1 /. {x->xy[[1]], y->xy[[2]]}}]**


**In[82] := P1 = ParametricPlot3D[xyz[t], {t, 0, 4},**

**PlotRange->{{-1, 0}, {-1, 0}, {-1, 0}}, ViewPoint->{10, 11, 12}];**

**In[83] := p0 = {-1/3, -1/3, -1/3};**

```
In[84] := P2 = Graphics3D[{Darker[Green], PointSize[Large], Point[p0]};
In[85] := Show[P1, P2]
```

Out[85] =



You can rotate this plot with your mouse to understand it better.

## 24.6 Shape of the Molecule

What does the cyclohexane molecule look like? Let's direct the  $x$ -axis along  $a[1]$ :

```
In[86] := a[1] = {1, 0, 0}
```

```
Out[86] = {1, 0, 0}
```

Let  $a[2]$  lie in the  $x, y$  plane:

```
In[87] := a[2] = {1/3, 2* Sqrt[2]/3, 0}
```

```
Out[87] = { 1/3, 2* Sqrt[2]/3, 0 }
```

That is, the unit vector along  $y$  is a combination of  $a[1]$  and  $a[2]$ :

```
In[88] := (3 * a[2] - a[1]) / (2 * Sqrt[2])
```

```
Out[88] = {0, 1, 0}
```

The projections of  $a[3]$  onto  $x$  and  $y$  are  $c[1, 3] = x$  and

```
In[89] := (3 * c[2, 3] - c[1, 3]) / (2 * Sqrt[2])
```

```
Out[89] = (1 - x) / (2 * Sqrt[2])
```

The projection of  $a[3]$  onto the  $z$  axis can be found from normalization:

```
In[90] := a[3] = {x, (1 - x)/(2 * Sqrt[2]),
  pm * Sqrt[(1 - x) * (7 + 9 * x)]/(2 * Sqrt[2])}
```

```
Out[90] = {x, 1 - x / (2 * Sqrt[2]), pm * Sqrt[(1 - x) * (7 + 9 * x)] / (2 * Sqrt[2])}
```

where  $pm = \pm 1$ . Two molecule shapes correspond to a single set of values of  $x, y, z$ ; they differ by the mirror reflection of the  $z$  coordinates. We shall discuss this matter in a moment.

```
In[91] := Table[Expand[a[i].a[3]]/.pm^2->1, {i, 1, 3}]
```

```
Out[91] = {x, 1/3, 1}
```

That is, the unit vector along the  $z$  axis is a combination of  $a[1], a[2], a[3]$ :

```
In[92] := Simplify[2 * Sqrt[2]/Sqrt[(1 - x) * (7 + 9 * x)] *
  (a[3] - 3/8 * (1 - x) * a[2] + (1 - 9 * x)/8 * a[1])]
```

```
Out[92] = {0, 0, pm}
```

The rest is easy.

```
In[93] := Do[Print[a[i] = Simplify[{c[1, i], (3 * c[2, i] - c[1, i])/(2 * Sqrt[2]),
  2 * Sqrt[2] * pm/Sqrt[(1 - x) * (7 + 9 * x)] *
  (c[3, i] - 3/8 * (1 - x) * c[2, i] + (1 - 9 * x)/8 * c[1, i])}],
```

```
{i, 4, 6}]
```

```
{-5/3 - x - z, 5/3 + x + 4z, pm(1 + 9x^2 - 4z + 2x(7 + 6z)) / (2 * Sqrt[2] * Sqrt[7 + 2x - 9x^2])},
{z, -5 + 3y + 4z, -pm(-5 - 11y - 4z + x(5 + 3y + 12z)) / (2 * Sqrt[2] * Sqrt[7 + 2x - 9x^2])},
{1/3, -1 + 9y, pm(-13 - 11y + x(-11 + 3y)) / (6 * Sqrt[2] * Sqrt[7 + 2x - 9x^2])}
```

Let's write a function which constructs the molecule for a given values of  $x, y, z$  and of the sign  $pm$ .


```
In[94] := rc = 0.1; rs = 0.25;
```

```
In[95] := Molecule[xyz_., s_] := Module[{r = {0, 0, 0}, r2, l = {Blue},
  S = {x->xyz[[1]], y->xyz[[2]], z->xyz[[3]], pm->s}},
  Do[r2 = r + (a[i]/.S); l = Append[l, Cylinder[{r, r2}, rc]]; r = r2, {i, 1, 6}];
  r = {0, 0, 0}; l = Append[l, Red];
  Do[r2 = r + (a[i]/.S); l = Append[l, Sphere[r, rs]]; r = r2, {i, 1, 6}];
  Graphics3D[l]]
```

This is the isolated conformation of the cyclohexane molecule with  $x = y = z = -1/3$ . Use your mouse to understand it better.

```
In[96] := Show[Molecule[p0, 1], ViewPoint->{15, -5, 5}, Boxed->False]
```

Out[96] =



And this is the one-parameter family of conformations. To ensure smooth dependence on  $t$ , it is necessary to flip the sign  $pm$  when passing through the point C (where the expression under the radical sign vanishes). This happens at

**In[97] :=  $t_0 = (621 - 8 * \text{Sqrt}[6])/159$**

**Out[97] =  $\frac{1}{159} (621 - 8\sqrt{6})$**

So, the molecule returns to its initial shape after we traverse the loop in the  $x$ ,  $y$ ,  $z$  space twice. You can see this conformation family especially clearly if you start animation.

**In[98] := Manipulate[Show[  
 Molecule[If[ $t > 4$ , xyz[ $t - 4$ ], xyz[ $t$ ]], If[ $t > t_0 \&\& t < t_0 + 4$ , -1, +1]],  
 PlotRange -> {{-0.7, 1.7}, {-0.5, 1.9}, {-1.7, 1.7}},  
 ViewPoint -> {10, -10, 4}, Boxed -> False],  
 { $t$ , 0, 8}]**

Out[98] =

